MATRICES CON MATHEMATICA

Dada la matriz $A = \begin{pmatrix} 1 & m & m^2 & m^3 \\ m & m^2 & m^3 & 1 \\ m^2 & m^3 & 1 & m \\ m^3 & 1 & m & m^2 \end{pmatrix}$, **hallar el valor del número natural m que hace que el determinante de la matriz A sea** 80^3

```
Clear["Global`*"]
```

$$a = \begin{pmatrix} 1 & m & m^2 & m^3 \\ m & m^2 & m^3 & 1 \\ m^2 & m^3 & 1 & m \\ m^3 & 1 & m & m^2 \end{pmatrix};$$

```
Solve[Det[a] == 80^3, m]
```

$$\left\{\{m \to -3\}, \{m \to -3\,i\}, \{m \to 3\,i\}, \{m \to 3\}, \left\{m \to -\left(-39 - 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to -i\left(-39 - 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to i\left(-39 - 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to \left(-39 - 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to -\left(-39 + 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to -i\left(-39 + 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to i\left(-39 + 40\,i\sqrt{3}\right)^{1/4}\right\}, \left\{m \to \left(-39 + 40\,i\sqrt{3}\right)^{1/4}\right\}\right\}$$

```
(* El único número natural es m=3 *)
```

Hallar una matriz que sumada a la matriz $\begin{pmatrix} 3 & 0 & 1 \\ -2 & 3 & 7 \end{pmatrix}$ **de la matriz** $\begin{pmatrix} -4 & 3 & 2 \\ 1 & 1 & 0 \end{pmatrix}$

```
a = {{3, 0, 1}, {-2, 3, 7}};
```

```
b = {{-4, 3, 2}, {1, 1, 0}};
```

```
c = Array[x, {2, 3}]
```

```
{{x[1, 1], x[1, 2], x[1, 3]}, {x[2, 1], x[2, 2], x[2, 3]}}
```

```
s = Solve[a + c == b]
```

```
{{x[1, 1] → -7, x[1, 2] → 3, x[1, 3] → 1, x[2, 1] → 3, x[2, 2] → -2, x[2, 3] → -7}}
```

```
c /. Flatten[s]
```

```
{{-7, 3, 1}, {3, -2, -7}}
```

```
MatrixForm[%]
```

$$\begin{pmatrix} -7 & 3 & 1 \\ 3 & -2 & -7 \end{pmatrix}$$

Partiendo de la matriz $\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$ extraer la matriz $\begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}$

```
Clear["Global`*"]
```

```
a = {{1, 2, 3}, {4, 5, 6}, {7, 8, 9}};
```

```
MatrixForm[a]
```

$$\begin{pmatrix} 1 & 2 & 3 \\ 4 & 5 & 6 \\ 7 & 8 & 9 \end{pmatrix}$$

Método 1

```
b = Drop[a, -1]
```

```
{{1, 2, 3}, {4, 5, 6}}
```

```
c = Transpose[b]
```

```
{{1, 4}, {2, 5}, {3, 6}}
```

```
d = Drop[c, -1]
```

```
{{1, 4}, {2, 5}}
```

```
Transpose[d]
```

```
{{1, 2}, {4, 5}}
```

Método 2

```
a[[{1, 2}, {1, 2}]]
```

```
{{1, 2}, {4, 5}}
```

```
MatrixForm[%]
```

$$\begin{pmatrix} 1 & 2 \\ 4 & 5 \end{pmatrix}$$

Estudiar el rango de la matriz $\begin{pmatrix} 1 & a & 1 \\ 1 & -2 & 0 \\ 2a & 1 & a \end{pmatrix}$

```
Clear["Global`*"]
```

$$\text{mat} = \begin{pmatrix} 1 & a & 1 \\ 1 & -2 & 0 \\ 2a & 1 & a \end{pmatrix};$$

Método 1

```
Solve[Det[mat] == 0, a]
```

$$\left\{\left\{a \to 1 - \sqrt{2}\right\}, \left\{a \to 1 + \sqrt{2}\right\}\right\}$$

(* si a≠ $1-\sqrt{2}$ y a≠ $1+\sqrt{2}$ →
el rango de la matriz es 3 ya que el determinante de orden 3 es distinto de 0 *)

```
d = mat[[{1, 2}, {1, 3}]]
```

```
{{1, 1}, {1, 0}}
```

```
Det[d]
```

```
-1
```

```
(* determinante de orden 2 ≠0 → si a=
 1-√2 o a= 1+√2 el rango de la matriz es 2 *)
```

Método 2

```
Minors[mat]
```

$$\left\{\{-2-a,\ -1,\ 2\},\ \{1-2a^2,\ -a,\ -1+a^2\},\ \{1+4a,\ a,\ -2a\}\right\}$$

```
(* hay varios menores de orden 2
   distinto de 0 que no dependen del valor del parámetro →
  si a= 1-√2 o a= 1+√2 el rango de la matriz es 2 *)
```

Dada la matriz $A = \begin{pmatrix} 1 & 2 \\ 3 & m \end{pmatrix}$,

calcular el valor del parámetro real m de modo que existan matrices no nulas que cumplan A.B = O.

Indicar la matriz A y la expresión general de B

```
Clear["Global`*"]
```

```
a = {{1, 2}, {3, m}};
```

```
b = Array[x, {2, 2}];
```

```
s = Solve[a.b == {{0, 0}, {0, 0}}, m]
```

```
{{m → 6}}
```

```
a = a /. Flatten[s]
```

```
{{1, 2}, {3, 6}}
```

```
s = Solve[a.b == {{0, 0}, {0, 0}}, Flatten[b]]
```

Solve::svars : Equations may not give solutions for all "solve" variables. >>

```
{{x[1, 1] → -2 x[2, 1], x[1, 2] → -2 x[2, 2]}}
```

```
b = b /. Flatten[s]
```

```
{{-2 x[2, 1], -2 x[2, 2]}, {x[2, 1], x[2, 2]}}
```

```
MatrixForm[b]
```

$$\begin{pmatrix} -2\,x[2,1] & -2\,x[2,2] \\ x[2,1] & x[2,2] \end{pmatrix}$$

Dada la matriz $\begin{pmatrix} x & x+y & x & x \\ x & 2x & x & x \\ x & x+y & y & x \\ y & x+y & x & x \end{pmatrix}$

a) Calcular su determinante

b) Indicar en que filas y columnas aparece el elemento x + y

c) Especificar las diagonales paralelas a la doagonal principal

```
a = {{x, x + y, x, x}, {x, 2 x, x, x}, {x, x + y, y, x}, {y, x + y, x, x}};
```

```
Det[a] // Simplify
```

$-x\,(x-y)^3$

```
Position[a, x + y]
```

```
{{1, 2}, {3, 2}, {4, 2}}
```

```
Table[Diagonal[a, n], {n, -3, 3}]
```

```
{{y}, {x, x + y}, {x, x + y, x}, {x, 2 x, y, x}, {x + y, x, x}, {x, x}, {x}}
```

Sea la matriz $\begin{pmatrix} 3 & 2 & 1 & -1 & 5 \\ 0 & 3 & 2 & 0 & 0 \\ 1 & 2 & 1 & 3 & 1 \\ 0 & 1 & -1 & 2 & 1 \\ 3 & 0 & 1 & 2 & 1 \end{pmatrix}$**, hallar el menor complementario del elemento situado en la fila 2, columna 3.**

```
a = {{3, 2, 1, -1, 5}, {0, 3, 2, 0, 0}, {1, 2, 1, 3, 1}, {0, 1, -1, 2, 1}, {3, 0, 1, 2, 1}};
```

```
MatrixForm[a]
```

$$\begin{pmatrix} 3 & 2 & 1 & -1 & 5 \\ 0 & 3 & 2 & 0 & 0 \\ 1 & 2 & 1 & 3 & 1 \\ 0 & 1 & -1 & 2 & 1 \\ 3 & 0 & 1 & 2 & 1 \end{pmatrix}$$

```
b = a[[{1, 3, 4, 5}, {1, 2, 4, 5}]]
```

```
{{3, 2, -1, 5}, {1, 2, 3, 1}, {0, 1, 2, 1}, {3, 0, 2, 1}}
```

```
MatrixForm[b]
```

$$\begin{pmatrix} 3 & 2 & -1 & 5 \\ 1 & 2 & 3 & 1 \\ 0 & 1 & 2 & 1 \\ 3 & 0 & 2 & 1 \end{pmatrix}$$

```
menor = Det[b]
```

```
-38
```

Dada la matriz A= $\begin{pmatrix} 1 & x & x & x & x \\ 2 & 1 & 2x & 2x & 2x \\ 2 & 2 & 1 & 2x & 2x \\ 2 & 2 & 2 & 1 & 2x \\ 2 & 2 & 2 & 2 & 1 \end{pmatrix}$ **, hallar el valor de x de modo que A sea regular.Calcular y comprobar que efectivamente es la inversa de A.**

```
a = {{1, x, x, x, x}, {2, 1, 2 x, 2 x, 2 x},
    {2, 2, 1, 2 x, 2 x}, {2, 2, 2, 1, 2 x}, {2, 2, 2, 2, 1}};
```

```
Solve[Det[a] == 0, x]
```

$$\left\{\left\{x \to \frac{1}{2}\right\}, \left\{x \to \frac{1}{2}\right\}, \left\{x \to \frac{1}{2}\right\}, \left\{x \to \frac{1}{2}\right\}\right\}$$

(* la matriz es regular cuando $x \neq \frac{1}{2}$ *)

```
b = Inverse[a] // Simplify
```

$$\left\{\left\{\frac{1-8x+16x^2-16x^3}{(1-2x)^4}, \frac{x}{(1-2x)^4}, \frac{x}{(-1+2x)^3}, \frac{x}{(1-2x)^2}, \frac{x}{-1+2x}\right\}, \left\{\frac{2}{-1+2x}, \frac{1}{1-2x}, 0, 0, 0\right\}, \left\{\frac{2}{(1-2x)^2}, \frac{2(-1+x)}{(1-2x)^2}, \frac{1}{1-2x}, 0, 0\right\}, \left\{\frac{2}{(-1+2x)^3}, \frac{2(-1+x)}{(-1+2x)^3}, \frac{2(-1+x)}{(1-2x)^2}, \frac{1}{1-2x}, 0\right\}, \left\{\frac{2}{(1-2x)^4}, \frac{2(-1+x)}{(1-2x)^4}, \frac{2(-1+x)}{(-1+2x)^3}, \frac{2(-1+x)}{(1-2x)^2}, \frac{1}{1-2x}\right\}\right\}$$

```
a.b // Simplify
```

```
{{1, 0, 0, 0, 0}, {0, 1, 0, 0, 0}, {0, 0, 1, 0, 0}, {0, 0, 0, 1, 0}, {0, 0, 0, 0, 1}}
```

```
MatrixForm[%]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

```
(* comprobación *)
```

```
a.b == IdentityMatrix[5] // Simplify
```

```
True
```

Sean las matrices A y B que cumplen que $A + B = \begin{pmatrix} 3 & 2 \\ 7 & 0 \end{pmatrix}$ y $A - B = \begin{pmatrix} 2 & 3 \\ -1 & 0 \end{pmatrix}$, calcular $A^2 + B^2$

```
a = Array[x, {2, 2}]; b = Array[y, {2, 2}];
```

```
s = Solve[{a + b == {{3, 2}, {7, 0}}, a - b == {{2, 3}, {-1, 0}}}]
```

$$\left\{\left\{x[1,1] \to \frac{5}{2}, y[1,1] \to \frac{1}{2}, x[1,2] \to \frac{5}{2}, y[1,2] \to -\frac{1}{2}, x[2,1] \to 3, y[2,1] \to 4, x[2,2] \to 0, y[2,2] \to 0\right\}\right\}$$

```
a = a /. Flatten[s]
```

$\left\{\left\{\frac{5}{2}, \frac{5}{2}\right\}, \{3, 0\}\right\}$

```
b = b /. Flatten[s]
```

$\left\{\left\{\frac{1}{2}, -\frac{1}{2}\right\}, \{4, 0\}\right\}$

```
MatrixPower[a, 2] + MatrixPower[b, 2]
```

$\left\{\{12, 6\}, \left\{\frac{19}{2}, \frac{11}{2}\right\}\right\}$

Dadas las matrices A=$\begin{pmatrix} 2 & 3 \\ 4 & 1 \end{pmatrix}$, B=$\begin{pmatrix} 3 & 2 \\ 1 & 0 \end{pmatrix}$, C=$\begin{pmatrix} 2 & 1 \\ 2 & 0 \end{pmatrix}$ y D=$\begin{pmatrix} 3 & 2 \\ 1 & 0 \end{pmatrix}$, y calcular las matrices M y N que resuelven el siguiente sistema de ecuaciones matriciales: $\begin{cases} M^T + AN = B \\ M + N^T C = D \end{cases}$

```
a = {{2, 3}, {4, 1}}; b = {{3, 2}, {1, 0}}; c = {{2, 1}, {2, 0}}; d = {{3, 2}, {1, 0}};
```

```
m = Array[x, {2, 2}]; n = Array[y, {2, 2}];
```

```
s = Solve[{Transpose[m] + a.n == b, m + Transpose[n].c == d}]
```

$\left\{\left\{x[1,2] \to \frac{7}{3}, y[1,1] \to -\frac{1}{3}, x[2,2] \to \frac{1}{3},\right.\right.$
$\left.\left. y[1,2] \to -\frac{1}{3}, y[2,1] \to 0, x[1,1] \to \frac{11}{3}, y[2,2] \to 1, x[2,1] \to -\frac{1}{3}\right\}\right\}$

```
m = m /. Flatten[s]
```

$\left\{\left\{\frac{11}{3}, \frac{7}{3}\right\}, \left\{-\frac{1}{3}, \frac{1}{3}\right\}\right\}$

```
MatrixForm[m]
```

$\begin{pmatrix} \frac{11}{3} & \frac{7}{3} \\ -\frac{1}{3} & \frac{1}{3} \end{pmatrix}$

```
n = n /. Flatten[s]
```

$$\left\{\left\{-\frac{1}{3}, -\frac{1}{3}\right\}, \{0, 1\}\right\}$$

```
MatrixForm[n]
```

$$\begin{pmatrix} -\frac{1}{3} & -\frac{1}{3} \\ 0 & 1 \end{pmatrix}$$

Calcular la matriz inversa de la matriz A= $\begin{pmatrix} 1 & 1 & 1 & 1 \\ 1 & 1 & 3 & 4 \\ 1 & 1 & 2 & 3 \\ 2 & -1 & 1 & 4 \end{pmatrix}$ **aplicando el método de Gauss-Jordan**

Las operaciones elementales que se van a aplicar son:

- Función Pivotar: Calcula la matriz que resulta de realizar sobre las filas de A la operación elemental que transforma el elemento a_{ij} en cero
- Función Intercambio: Calcula la matriz que resulta de realizar sobre las filas de A la operación elemental de intercambiar entre si dos filas de A
- Función Proporcional: Calcula la matriz que resulta de realizar sobre las filas de A la operación elemental que de multiplicar una fila (i-ésima) por un número "x"
- Función Múltiplo: Calcula el valor del parámetro "x" en la operación elemental $F_i \Longleftarrow F_i + x.F_j$ que hace nulo el elemento a_{ij} de la matriz A

```
Clear["Global`*"]
```

```
Pivotar[m_, i_, j_] := ReplacePart[m, m[[i]] - (m[[i, j]] / m[[j, j]]) m[[j]], i]
Intercambio[m_, i_, j_] := ReplacePart[ReplacePart[m, m[[j]], i], m[[i]], j]
Proporcional[m_, i_, lambda_] := ReplacePart[m, m[[i]] * lambda, i]
Multiplo[m_, i_, j_] := - (m[[i, j]] / m[[j, j]])
```

Se introduce la matriz ampliada ⟦ A | I ⟧ , que se denomina MatAI

```
MatAI = {{1, 1, 1, 1, 1, 0, 0, 0}, {1, 1, 3, 4, 0, 1, 0, 0},
   {1, 1, 2, 3, 0, 0, 1, 0}, {2, -1, 1, 4, 0, 0, 0, 1}};
```

```
MatAI // MatrixForm
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 1 & 1 & 3 & 4 & 0 & 1 & 0 & 0 \\ 1 & 1 & 2 & 3 & 0 & 0 & 1 & 0 \\ 2 & -1 & 1 & 4 & 0 & 0 & 0 & 1 \end{pmatrix}$$

(* **PRIMERA ETAPA** *)

(* Utilizando la función pivotar se generan elementos nulos por debajo de la diagonal principal. *)
(* Generación de ceros en la primera columna por debajo de la diagonal principal. Se hacen ceros los elementos a_{21} , a_{31}, a_{41}. El resultado
es una nueva matyriz simbolizada por ⟦ A | I ⟧ quie se ve con el comando MatrixForm *)

```
MatrixForm[MatAI = Pivotar[MatAI, 2, 1]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 3 & -1 & 1 & 0 & 0 \\ 1 & 1 & 2 & 3 & 0 & 0 & 1 & 0 \\ 2 & -1 & 1 & 4 & 0 & 0 & 0 & 1 \end{pmatrix}$$

```
MatrixForm[MatAI = Pivotar[MatAI, 3, 1]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 3 & -1 & 1 & 0 & 0 \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 2 & -1 & 1 & 4 & 0 & 0 & 0 & 1 \end{pmatrix}$$

```
MatrixForm[MatAI = Pivotar[MatAI, 4, 1]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & 0 & 2 & 3 & -1 & 1 & 0 & 0 \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & -3 & -1 & 2 & -2 & 0 & 0 & 1 \end{pmatrix}$$

(* Para generar ceros en la segunda columna y dado que el elemento pivote es $a_{22} = 0$, Se busca un elemento en la segunda columna ,
debajo del a_{22}que sea distinto de cero. El elemento $a_{42} = -3 \neq 0$. Se intercambian las filas 2 y 4 entre si, utilizando la función Intercambio *)

```
MatrixForm[MatAI = Intercambio[MatAI, 2, 4]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & -3 & -1 & 2 & -2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 2 & 3 & -1 & 1 & 0 & 0 \end{pmatrix}$$

(* Se hace cero el elemento a_{43} *)

```
MatrixForm[MatAI = Pivotar[MatAI, 4, 3]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & -3 & -1 & 2 & -2 & 0 & 0 & 1 \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 & 1 & 1 & -2 & 0 \end{pmatrix}$$

(* **SEGUNDA ETAPA** *)

(* La función Proporcional transforma en unos los elementos de la diagonal principal a_{22} y a_{44}. *)

```
MatrixForm[MatAI = Proporcional[MatAI, 2, -1/3]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & \frac{1}{3} & -\frac{2}{3} & \frac{2}{3} & 0 & 0 & -\frac{1}{3} \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & -1 & 1 & 1 & -2 & 0 \end{pmatrix}$$

```
MatrixForm[MatAI = Proporcional[MatAI, 4, -1]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 1 & 1 & 0 & 0 & 0 \\ 0 & 1 & \frac{1}{3} & -\frac{2}{3} & \frac{2}{3} & 0 & 0 & -\frac{1}{3} \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

(* **TERCERA ETAPA** *)

(* Hacer nulos los elementos situados por encima de la diagonal principal. Se comienza por la 4ª columna: elementos a_{14}, a_{24}, a_{34} utilizando la función Pivotar *)

```
MatrixForm[MatAI = Pivotar[MatAI, 1, 4]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 0 & 2 & 1 & -2 & 0 \\ 0 & 1 & \frac{1}{3} & -\frac{2}{3} & \frac{2}{3} & 0 & 0 & -\frac{1}{3} \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

```
MatrixForm[MatAI = Pivotar[MatAI, 2, 4]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 0 & 2 & 1 & -2 & 0 \\ 0 & 1 & \frac{1}{3} & 0 & 0 & -\frac{2}{3} & \frac{4}{3} & -\frac{1}{3} \\ 0 & 0 & 1 & 2 & -1 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

```
MatrixForm[MatAI = Pivotar[MatAI, 3, 4]]
```

$$\begin{pmatrix} 1 & 1 & 1 & 0 & 2 & 1 & -2 & 0 \\ 0 & 1 & \frac{1}{3} & 0 & 0 & -\frac{2}{3} & \frac{4}{3} & -\frac{1}{3} \\ 0 & 0 & 1 & 0 & 1 & 2 & -3 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

(* Hacer nulos los elementos situados por encima de la diagonal principal. Se continúa por la 3ª columna: elementos a_{13}, a_{23} utilizando la función Pivotar *)

```
MatrixForm[MatAI = Pivotar[MatAI, 1, 3]]
```

$$\begin{pmatrix} 1 & 1 & 0 & 0 & 1 & -1 & 1 & 0 \\ 0 & 1 & \frac{1}{3} & 0 & 0 & -\frac{2}{3} & \frac{4}{3} & -\frac{1}{3} \\ 0 & 0 & 1 & 0 & 1 & 2 & -3 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

```
MatrixForm[MatAI = Pivotar[MatAI, 2, 3]]
```

$$\begin{pmatrix} 1 & 1 & 0 & 0 & 1 & -1 & 1 & 0 \\ 0 & 1 & 0 & 0 & -\frac{1}{3} & -\frac{4}{3} & \frac{7}{3} & -\frac{1}{3} \\ 0 & 0 & 1 & 0 & 1 & 2 & -3 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

(* Hacer nulos los elementos situados por encima de la diagonal principal. Se continúa por la 2ª columna: elemento a_{12} , utilizando la función Pivotar *)

```
MatrixForm[MatAI = Pivotar[MatAI, 1, 2]]
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 & \frac{4}{3} & \frac{1}{3} & -\frac{4}{3} & \frac{1}{3} \\ 0 & 1 & 0 & 0 & -\frac{1}{3} & -\frac{4}{3} & \frac{7}{3} & -\frac{1}{3} \\ 0 & 0 & 1 & 0 & 1 & 2 & -3 & 0 \\ 0 & 0 & 0 & 1 & -1 & -1 & 2 & 0 \end{pmatrix}$$

(* El primer bloque (matriz $A_{(4)}$ en la que se ha operado sde ha transformado en la matriz identidad $I_{(4)}$, luego la matriz del segundo bloque formado por las columnas 5ª, 6ª, 7ª y 8ª es la matriz inversa de A → $A^{-1} = \begin{pmatrix} \frac{4}{3} & \frac{1}{3} & -\frac{4}{3} & \frac{1}{3} \\ -\frac{1}{3} & -\frac{4}{3} & \frac{7}{3} & -\frac{1}{3} \\ 1 & 2 & -3 & 0 \\ -1 & -1 & 2 & 0 \end{pmatrix}$ *)